

# An Ontology for Describing Cosmological Objects

F. Sáenz-Pérez  
GPD-DISIA-UCM  
April 2012

## 1 Goals

- Implement an ontology following [AVS08] for describing cosmological objects
- Represent concepts, relations, and algebraic properties attached to relations
- Enable constraint checking, both hard and soft. Hard constraint checking forbids inconsistent databases, whereas soft constraint checking allows them

## 2 Application Description

Cosmology deals with a number of objects that can be represented in a hierarchy [Nar96]. Figure 1 represents a simplified structure of the universe, which is composed of galaxies, and each galaxy is composed of other objects, as planetary systems, each one, in turn, composed of stars, planets and satellites. Both galaxies, stars, planets and satellites are known as cosmological objects (either simply objects or bodies; both terms are interchangeably used). There are objects that are components of bigger ones or even which are not, as, e.g., a star and the universe itself, respectively. In addition, there are objects that orbit others, as the Moon, which orbits the Earth. For an object to be considered as a planet, it must directly orbit a star, and no intermediate orbiting object can be found in-between. Analogously, a satellite must directly orbit a planet. Also, for an object to be considered as a planetary system, at least it must include a planet. From now on, we distinguish between "type of cosmological object" and "cosmological object", as "planet" and "Earth", respectively. Then, an object in any cosmological instance must be either a "type of cosmological object" or a "cosmological object", but not both at the same time.

Several semantic relations naturally emerge from this description: "component of", "is a", and "orbits", all of which relate concepts. The basic concepts are the cosmological objects, as "star" and "sun" (we say that the Sun is a star). The basic relation in this application is "component of", to which the algebraic properties "irreflexive", "asymmetric" and "transitive" are attached. In turn, both "is a" and "orbit" enjoy the properties "reflexive", "asymmetric" and "transitive".

## 3 Application Implementation

We use DES [Sae11] as a host system to implement the ontology, which supports strong constraints, constraint checking, and consistency checking. Datalog program examples cited in this paper can be found in `cosmos.dl`.

### 3.1 Data Modelling

We focus on the part of the E-R model subpart depicted in Fig. 1 relating "Concepts", "Relations", "BinaryRelation", "HasAP", and "AlgebraicProperty"




Fig. 1. Ontology Metadata E-R Model

Translating this conceptual model into the relational model, we get:

```

concepts(Type, Id, Arity)
Type ∈ {concept, relation}
Arity ≥ 0

```

```

binary_relation(Id, FromId, ToId)
binary_relation.FromId ∈ concepts.Id
binary_relation.ToId ∈ concepts.Id

```

```

algebraic_properties(Id)

```

```

intrinsic_properties(Id)

```

```

has_algebraic_property(BinaryRelation, AlgebraicProperty)
has_algebraic_property.BinaryRelation ∈ binary_relation.Id
has_algebraic_property.AlgebraicProperty ∈ algebraic_properties.Id

```

```

has_intrinsic_property(Concept, IntrinsicProperty)
has_intrinsic_property.Concept ∈ concepts.Id
has_intrinsic_property.IntrinsicProperty ∈ intrinsic_properties.Id

```

The complete physical model can be examined in the file "**cosmos.d1**".

A possible instance of this ontology is:

**% Concepts:**

```

% - Objects:
concepts(concept, object_type, 0).
concepts(concept, object_instance, 0).
% - Object types:
concepts(concept, universe, 0).
concepts(concept, galaxy, 0).
concepts(concept, planetary_system, 0).
concepts(concept, star, 0).
concepts(concept, planet, 0).
concepts(concept, satellite, 0).
% - Object instances:
concepts(concept, 'Universe', 0).
concepts(concept, 'Milky Way', 0).
concepts(concept, 'Solar System', 0).
concepts(concept, 'Sun', 0).
concepts(concept, 'Earth', 0).
concepts(concept, 'Moon', 0).
% Relations:
concepts(relation, component_of, 2).
concepts(relation, is_a, 2).
concepts(relation, orbits, 2).

binary_relation(is_a, universe, object_type).
binary_relation(is_a, galaxy, object_type).
binary_relation(is_a, planetary_system, object_type).
binary_relation(is_a, star, object_type).
binary_relation(is_a, planet, object_type).
binary_relation(is_a, satellite, object_type).

binary_relation(is_a, 'Universe', object_instance).
binary_relation(is_a, 'Milky Way', object_instance).
binary_relation(is_a, 'Solar System', object_instance).
binary_relation(is_a, 'Sun', object_instance).
binary_relation(is_a, 'Earth', object_instance).
binary_relation(is_a, 'Moon', object_instance).

binary_relation(is_a, 'Universe', universe).
binary_relation(is_a, 'Milky Way', galaxy).
binary_relation(is_a, 'Solar System', planetary_system).
binary_relation(is_a, 'Sun', star).
binary_relation(is_a, 'Earth', planet).
binary_relation(is_a, 'Moon', satellite).

binary_relation(component_of, galaxy, universe).
binary_relation(component_of, planetary_system, galaxy).
binary_relation(component_of, star, planetary_system).
binary_relation(component_of, planet, planetary_system).
binary_relation(component_of, satellite, planetary_system).

% binary_relation(component_of, 'Milky Way', 'Universe').
% binary_relation(component_of, 'Solar System', 'Milky Way').
% binary_relation(component_of, 'Sun', 'Solar System').
% binary_relation(component_of, 'Earth', 'Solar System').
% binary_relation(component_of, 'Moon', 'Solar System').

algebraic_properties(reflexive).

```

```

algebraic_properties(irreflexive).
algebraic_properties(asymmetric).
algebraic_properties(transitive).

intrinsic_properties(is_planet).

has_algebraic_property(is_a, reflexive).
has_algebraic_property(is_a, asymmetric).
has_algebraic_property(is_a, transitive).

has_algebraic_property(component_of, irreflexive).
has_algebraic_property(component_of, asymmetric).
has_algebraic_property(component_of, transitive).

has_algebraic_property(part_of, irreflexive).
has_algebraic_property(part_of, asymmetric).
has_algebraic_property(part_of, transitive).

has_intrinsic_property(planet, is_planet).

```

## 3.2 Constraints

Algebraic properties can be seen as constraints on a database. These properties, when viewed as constraints, can be classified as exclusion constraints and inclusion constraints. An *exclusion* constraint forbids in an instance database those values which are not consistent w.r.t. this constraint. An *inclusion* constraint specifies all the values that must be in an instance database for it to be consistent. Examples of exclusion constraints include irreflexive and asymmetric, whereas transitive and symmetric are examples of inclusion constraints.

Constraints can also be classified as hard and soft. A *hard* constraint on a database is checked whenever any change related to the involved data sources for the constraint occurs. A *soft* constraint is only checked on-demand. So, in this case, it may be the case that the database becomes inconsistent w.r.t. to the soft constraint. However, they have been acknowledged as a need for certain situations. For instance, in relational databases the concept of *deferred* constraint is used to denote such soft constraints.

### 3.2.1 Stating Hard Constraints

Attaching an exclusion constraint to a relation implies to impose a strong constraint, which is checked with a Datalog integrity constraint. As an example:

```
:- irreflexive_violation(BinaryRelation, From, To).
```

where:

```

irreflexive_violation(BinaryRelation, From, From) :-
    has_algebraic_property(BinaryRelation, irreflexive),
    binary_relation(BinaryRelation, From, From).

```

That is, it cannot be the case to find a binary relation relating the same element with itself.

Following the example in Section 3.1, trying to assert an entry which makes the ontology inconsistent w.r.t. this algebraic property is rejected:

```
DES> /assert binary_relation(component_of, planet, planet).
```

```
Error: Integrity constraint violation.
```

```
ic(BinaryRelation,From,To) :-
```

```
    irreflexive_violation(BinaryRelation,From,To).
```

```
Offending values in database:
```

```
[ic(component_of,planet,planet)]
```

Attaching an inclusion constraint to a relation as a strong constraint implies in general to intensionally represent the meaning of the relation under such constraint. Otherwise, there are situations for which it is not possible to assert new data although it is possible to reach a consistent instance database eventually, after further assertions.

For example, considering the property "**transitive**", we can have the following instance database:

```
binary_relation(spaceflight, 'Earth', 'Mars').
```

```
binary_relation(spaceflight, 'Earth', 'Moon').
```

```
binary_relation(spaceflight, 'Mars', 'Moon').
```

And impose the property "**transitive**" as a hard constraint:

```
:- transitive_violation(BinaryRelation, From, To).
```

where:

```
transitive_violation(BinaryRelation, From, To) :-
```

```
    BinaryRelation = spaceflight,
```

```
    has_algebraic_property(BinaryRelation, transitive),
```

```
    count(binary_relation(spaceflight, _, _),RelationCardinality),
```

```
    count(transitive_spaceflight(BinaryRelation, _,
```

```
_),TCCardinality),
```

```
    RelationCardinality \= TCCardinality,
```

```
    transitive_spaceflight(BinaryRelation, From, To),
```

```
    not(binary_relation(spaceflight, From, To)).
```

```
transitive_spaceflight(BinaryRelation, From, To) :-
```

```
    has_algebraic_property(BinaryRelation, transitive),
```

```
    binary_relation(BinaryRelation, From, To).
```

```
transitive_spaceflight(BinaryRelation, From, To) :-
```

```
    has_algebraic_property(BinaryRelation, transitive),
```

```
    binary_relation(BinaryRelation, From, Mid),
```

```
    transitive_spaceflight(BinaryRelation, Mid, To).
```

Then, it is neither possible to add:

```
binary_relation(spaceflight, 'Moon', 'Earth').
```

nor:

```
binary_relation(spaceflight, 'Mars', 'Earth').
```

because the first tuple requires the second already asserted and the other way round (this example can be found in the file `spaceflight.dl`), as illustrated in the next system session:

```
DES> /assert binary_relation(spaceflight, 'Moon', 'Earth').
```

```
Error: Integrity constraint violation.
```

```
ic(BinaryRelation,From,To) :-
```

```
    transitive_violation(BinaryRelation,From,To).
```

```
Offending values in database:
```

```
[ic(spaceflight,'Mars','Earth'),ic(spaceflight,'Earth','Earth'),  
ic(spaceflight,'Moon','Mars'),ic(spaceflight,'Moon','Moon'),ic(s  
paceflight,'Mars','Mars')]
```

```
Error: Asserting rules due to integrity constraint violation.
```

```
DES> /assert binary_relation(spaceflight, 'Mars', 'Earth').
```

```
Error: Integrity constraint violation.
```

```
ic(BinaryRelation,From,To) :-
```

```
    transitive_violation(BinaryRelation,From,To).
```

```
Offending values in database:
```

```
[ic(spaceflight,'Earth','Earth'),ic(spaceflight,'Mars','Mars')]
```

```
Error: Asserting rules due to integrity constraint violation.
```

Therefore, one possibility is to consider the property "**transitive**" as a soft constraint (c.f. next section) and another one is to consider that the meaning of a relation is extended with the intensional meaning derived by the inclusion property. This way, there is no need for checking consistency w.r.t. to inclusion properties as it is ensured by the definition of the relation.

Following this example about the property "**transitive**", we can extend the meaning of relations with this property attached as follows:

```
binary_relation(BinaryRelation, From, To) :-  
    has_algebraic_property(BinaryRelation, transitive),  
    binary_relation(BinaryRelation, From, Mid),  
    binary_relation(BinaryRelation, Mid, To).
```

Now, any binary relation with attached transitive property, has its meaning extended with its transitive closure.

In the example, given the following explicit two tuples above for "spaceflight", already loaded in the database:

```
binary_relation(spaceflight,'Mars','Moon').  
binary_relation(spaceflight,'Moon','Earth').
```

Then, if we ask for its meaning, we get:

```
DES> binary_relation(spaceflight,From,To)
{
  binary_relation(spaceflight,'Mars','Earth'),
  binary_relation(spaceflight,'Mars','Moon'),
  binary_relation(spaceflight,'Moon','Earth')
}
Info: 3 tuples computed.
```

So that we see that the meaning of relations with the attached property transitive is automatically extended with its transitive closure.

We can see that incorrect tuples cannot be added in the presence of the transitive property. As "**component\_of**" has three properties attached (i.e., properties "**irreflexive**", "**asymmetric**", and "**transitive**"), then it is not possible to assert an entry for "**component\_of**" which implies a cycle in the directed graph that this relation forms. For instance:

```
DES> /assert binary_relation(component_of, galaxy, planet).

Error: Integrity constraint violation.
      ic(BinaryRelation,From,To) :-
        acyclic_violation(BinaryRelation,From,To).
      Offending values in database:
[ic(component_of,bike,bike),ic(component_of,bearing,bearing),ic(
component_of,wheel,wheel)]
```

This shows that there are paths from "**bike**" to itself, and also for "**bearing**" and "**wheel**" (all of them form a cyclic path, which are not allowed because of the combination of the three properties).

### 3.2.2 Stating Soft Constraints.

If we omit the strong constraints, we can get inconsistent ontologies along authoring. Let's consider the following case:

```
% Concepts
concepts(concept, bike, 0).
concepts(concept, wheel, 0).
concepts(concept, spoke, 0).
concepts(concept, bearing, 0).
% Relations
concepts(relation, component_of, 2).

binary_relation(component_of, bearing, wheel).
binary_relation(component_of, spoke, wheel).
binary_relation(component_of, wheel, bike).
```

where the algebraic property "**irreflexive**" is not attached to "**component\_of**".

Then, we can assert an incorrect entry w.r.t. "**irreflexive**":

```
DES> /assert binary_relation(component_of, bearing, bearing).
```

which is not rejected.

To test which data is inconsistent, we write:

```
DES> irreflexive_violation(BinaryRelation, From, To).  
  
{  
  irreflexive_violation(component_of, bearing, bearing)  
}  
Info: 1 tuple computed.
```

Then, what are soft constraints useful for? Along ontology development, one might allow for inconsistent databases for a number of reasons as, e.g.:

- Unknown data
- Not yet completed data

For the second one, we can think of the algebraic property "**transitive**" as introduced in the previous section. In this case, we require the meaning of relations to be explicit in our database. This means that if the following entries are in the database:

```
binary_relation(component_of, bearing, wheel).  
binary_relation(component_of, spoke, wheel).  
binary_relation(component_of, wheel, bike).
```

then, the following ones must also be in the database:

```
binary_relation(component_of, bearing, bike).  
binary_relation(component_of, spoke, bike).
```

So, to test what are the missing tuples in a database because of the inclusion property "**transitive**", it is attached as a soft constraint, i.e., it is NOT imposed (here, it is shown only as a program remark, i.e., a line preceded by the symbol "%"):

```
%:- transitive_violation(BinaryRelation, From, To).
```

(If it is already imposed, you can remove it with:

```
DES> /drop_ic :- transitive_violation(BinaryRelation, From, To).  
  
).
```

This allows us to incrementally build the ontology up to a point in which we think it is consistent and, therefore, to be sure of it. Then, we submit the following consistency check:

```
DES> transitive_violation(BinaryRelation, From, To)  
  
{  
  transitive_violation(component_of, bearing, bike),  
  transitive_violation(component_of, spoke, bike)  
}  
Info: 2 tuples computed.
```



Here, we see that "**component\_of**" is not consistent w.r.t. the algebraic property "**transitive**" because the entries shown are missing in the database. If they are asserted, the relation becomes a truly explicit transitive relation and therefore passes the consistency check:

```
DES> transitive_violation(BinaryRelation, From, To)
```

```
{
  transitive_violation(component_of,bearing,bike),
  transitive_violation(component_of,spoke,bike)
}
```

```
Info: 2 tuples computed.
```

```
DES> /assert binary_relation(component_of, bearing, bike)
```

```
DES> transitive_violation(BinaryRelation, From, To)
```

```
{
  transitive_violation(component_of,spoke,bike)
}
```

```
Info: 1 tuple computed.
```

```
DES> /assert binary_relation(component_of, spoke, bike)
```

```
DES> transitive_violation(BinaryRelation, From, To)
```

```
{
}
```

```
Info: 0 tuples computed.
```

## References

- [AVS08] F.J. Álvarez, A. Vaquero, and F. Sáenz-Pérez, "Conceptual Modeling of Ontology-based Linguistic Resources with a Focus on Semantic Relations", In Proceedings of the Sixth International Conference on Language Resources and Evaluation (LREC 2008), May, 2008.
- [Sae11] F. Sáenz-Pérez, "DES: A Deductive Database System", ENTCS 271, pp. 63-78, 2011. [des.sourceforge.net](http://des.sourceforge.net)
- [Nar96] Narlikar, Jayant V. (1996). *Elements of Cosmology*. Universities Press.